U.S. DEPARTMENT OF THE INTERIOR

U.S. GEOLOGICAL SURVEY

Geographic information system (GIS) procedure for preliminary delineation of debris-flow hazard areas from a digital terrain model, Madison County, Virginia

By

Russell H. Campbell[1] and Pete Chirico[2]

Documentation for and preliminary evaluation of the digital terrain data and GIS procedures applied to make the potential debris-flow hazards map of U.S. Geological Survey Miscellaneous Investigations Series Map I-2623B (Morgan and others, in press)

Open-File Report 99-336

[1]U.S. Geological Survey, ret.
c/o Department of Earth Sciences
University of California, Riverside
Riverside, CA 92521

[2]U.S. Geological Survey
MS 926A, National Center
Reston, VA 20192

CONTENTS

ILLUSTRATIONS

TABLE

GEOGRAPHIC INFORMATION SYSTEM (GIS) PROCEDURE FOR PRELIMINARY DELINEATION OF DEBRIS-FLOW HAZARD AREAS FROM A DIGITAL TERRAIN MODEL, MADISON COUNTY, VIRGINIA

by Russell H. Campbell and Pete Chirico

INTRODUCTION

Studies of debris-flow and flood effects of a severe storm in Madison County, Virginia, on June 27, 1995, have identified dynamic (rainfall and average depth of debris-flow surges) and static (slope, channel position and relative elevation) terrain factors that contribute to vulnerability to debris-flow hazards (Morgan and others, 1997; Morgan and others, in press). The spatial distribution of vulnerable areas, as defined by this combination of factors, has been delineated on a map using a geographic information system (GIS) to manipulate terrain models made from U.S. Geological Survey digital line graph (DLG) data for 7.5 min. (1:24,000-scale) quadrangle maps. The procedure yields a map that shows the areas in and adjacent to stream channels that would be affected by a debris-flow surge of user-specified depth, as approximated from a digital terrain model. It differs from other procedures for mapping potential debris-flow runout, which require user inputs that specify initial volumes (e.g., Iverson and others, 1998; Ellen and others, 1993).

The procedure is carried out as a single ARC/INFO[1] .aml (ARC macro language) program. The preliminary hazard map prepared by this procedure for the Madison County study area identifies most of the areas that were affected by debris flows during the June 27, 1995, storm. Most of the disparities between model-predicted hazard areas and the inventoried effects of that storm can be attributed to inaccuracies in the digital terrain model, particularly where slopes and stream gradients are flatter than about 14 degrees. In addition, some computational problems arise where stream flow is other than north-to-south because of the sequence with which calculations are carried out in ARC/INFO GRID.

Post-storm measurements of elevation, slope, aspect, curvature, colluvial thickness, and amount of rainfall were made in the Madison County area, especially at source areas and along debris-flow paths (Morgan and others, 1997; Morgan and others, in press). The measurements included the depth of flows that moved through channels (vertical distances between thalwegs and mudlines flanking the channels). Measured depths ranged from less than 1 meter to nearly 14 meters, but most were about 3 meters. Moreover, in many channels the surge depths were approximately constant for long reaches of first-, second-, and perhaps higher-order drainage lines. This observation implies (a) slurry volume was increased as surges moved downstream, and (b) the heights of surge fronts were limited by some mechanism that prevented added volume from "piling-up" indefinitely.

The amount and the spatial distribution of rainfall of the June 27, 1995, storm was the principal dynamic control for the spatial distribution of debris flows. The absence of debris flows in adjacent parts of the Blue Ridge having nearly identical static properties (elevation, slope,

---

[1]The use of the brand name of the commercial software program does not constitute endorsement by the U.S. Geological Survey.

aspect, curvature, colluvial thickness and lithologic character) is a consequence of lower rainfall rates and amounts in those areas (Wieczorek and others, in preparation). The model for assessing potential debris-flow hazards, therefore, assumes that the entire study area will be subjected to rainfall intensity and duration similar to that received by terrain within the 650 mm isohyet for the storm of June 27, 1995. This simple model also ignores variations in thickness and shear resistance of colluvium, as well as aspect and curvature, which are factors that are known or inferred to affect the susceptibility of hillslopes to rainfall-caused initiation of debris flows.

PROCEDURE AND MODEL

The GIS model was developed chiefly as a raster procedure using ARC/INFO GRID functions developed for use in modeling terrain and hydrologic flow (ESRI, 1991; ESRI, 1992). From U.S.G.S. contour and stream digital line graphs (DLG's) raster grids of the study area were prepared so that elevation, degree of slope steepness, and aspect were calculated for each 10 meter by 10 meter cell (Figure 1, Figure 2). From the elevation grid, derivative grids were made for flow direction, flow accumulation, stream net (shown as a sequence of contiguous cells on a background of NODATA cells), and (Strahler model) stream order (Figure 3). For gridded elevations derived from available contour digital line graphs, we found (after several trials) that a flow accumulation of 50 cells is suitable for approximating the upper ends of the inventoried debris-flow channels. The 50-cell flow accumulation also identifies a usable approximation of first-order drainage lines as picked by inspection from the contour map at 1:24,000-scale, although the upstream end cells of first-order drainage lines identified in this way are commonly somewhat down channel from the upper ends as picked by visual inspection of the contours. The 50-cell number was not selected to represent cells in which debris flows were initiated by the June 27, 1995, storm, but to approximate upper ends of drainage lines in which channelized slurry flow may have accumulated sufficient volume (and flow depth) so that cells adjacent to the thalweg cells may be inundated. Because the post-storm inventories showed that virtually all debris flows originated from cells having slopes of 26 degrees or steeper, and that deposition commonly began as surges reached channel gradients of about 14 degrees, the .aml includes an (somewhat ineffective) attempt to incorporate this information into model constraints that include slope; however, the product of the .aml still includes areas adjacent to some stream lines that head in basins where all slopes are gentler than 26 degrees -- in effect, erroneously showing a greater potential for hazard than is actually expected. Consequently, a final step was to edit the hazard map manually to remove drainages from watersheds that lack slopes of 26 degrees or greater.

The stream channels are depicted as contiguous strings of single cells (a 1-cell-wide thalweg) on a background of NODATA cells. Elevations for each thalweg cell are determined from the grid of elevations (Figure 4). Stream gradient can be measured as the slope of each thalweg cell. Three new grids are formed by allocating the thalweg elevations to adjacent cells for three selected distances. (The distances are selected on the basis of the densities of drainage lines of different orders to minimize overlap of the expansion of one drainage into adjacent drainage basins or sub-basins. In the Madison County terrain, we used 50 m, 30 m, and 10 m, respectively.) A surge depth of 3 meters was then added to the allocated thalweg elevations in each of the three grids. The result is 3 grids of contiguous cells, expanded from the thalweg grid, with elevations of the assigned surge depth, on a background of NODATA cells (Figures 5a, 5b and 5c). Subtracting the terrain elevations from the allocated surge elevations, constrained to

differences greater than 0, yields three grids showing surge depths on a background of NODATA cells (Figures 6a, 6b and 6c). The surge-depth grids can be constrained by stream order by using still another grid, created to allocate stream order designations to cells for the maximum of the previously selected distances (50 m in Madison County; Figure 7). Channel width commonly increases, and stream density commonly decreases, with increasing stream order and decreasing stream gradient. Therefore, it seems appropriate to use wider limits for the allocations of higher-order drainage lines (Figure 8). At their down-channel extremes, most of the debris flows entered flooding streams that rose to cover the distal ends of fans. Consequently, the flood hazard is mapped as overriding the debris-flow hazard in areas below the upper limits of mapped flood inundation in 1995 (Figure 9).

DATA INPUT TO MODEL

Five grids provide the information input to the .aml that generated the hazard map:

<elevation-grid>, a grid of elevations in meters at a 10-m spacing. U.S.G.S. contour DLG's for the 6 quadrangles that cover the study area were used to prepare line covers for the contours of each quadrangle. After converting the contour DLG's to ARC covers, the covers for the six adjoining 7.5' quadrangles that make up the entire study area were joined using the ARC **append** command, and edited along the individual quadrangle boundaries to insure against generating boundary artifacts in subsequent manipulations of the data. Next, the ARC command, **createtin,** was used to create a TIN surface from the attributed contour lines cover; the TIN was clipped to the study area boundaries; the ARC command, **tinarc**, was applied to convert the TIN into a polygon coverage with attributes for elevation, aspect azimuth, and slope; and the ARC command, **polygrid**, was used to prepare elevation, slope, and aspect grids at 10-meter spacing from the polygon cover. (Note: An initial attempt to use **tinlattice** to create the grids directly from the TIN failed to provide grids that registered exactly to the other coverages of the same map area -- possibly a result of operator error in defining the TIN boundary). The grid, *<elevation-grid>*, is the result of the first iteration of the GRID function, **fill**, on the initial elevation grid derived from the polygon cover (**fill** is a subroutine used to remove spurious sinks and peaks from elevation grids). (Examination of current *<elevation-grid>* for the Madison County study area indicates that it requires further editing to become truly depressionless.) The example, Figure 1, displays the elevation, aspect, and slope combined as a shaded relief map.

<slope-grid>, is an integer grid created from the floating point grid initially derived from the polygon cover of the TIN, which was categorized using the GRID function **slice** for the more rapid calculation of the aml. *<slope-grid>* cells are attributed in five slope categories: 0-14 degrees, 14-26 degrees, 26-34 degrees, 34-45 degrees, and greater than 45 degrees. The example, Figure 2, displays the selected intervals of slope.

<stream-order-grid>, a grid showing stream lines, attributed with the Strahler stream order of each link, was prepared from the elevation grid using the **flowdirection** and **flowaccumulation** GRID functions to create grids of flow direction and, subsequently, flow accumulation. A stream net was then delineated using a threshold of 50 cells for flow accumulation in stream order. That is, all cells that receive flow from fewer than 50

cells are assigned the value NODATA. (50 10–m cells = 5000 square meters in area, or one-half a hectare.) The 50-cell threshold was selected, after trials, to represent a first-order drainage that appears appropriate for the 1:24,000-scale contour maps in the study area. Stream orders were determined using the GRID function **streamorder** to operate on the stream net and flow accumulation grids. The example, Figure 3, illustrates the result of the **streamorder** function. Notice that it does not everywhere reproduce the assignment of order that would be made by inspection of the contours.

<*thalweg-elevation-grid*>, a grid in which the stream net cells are attributed with their elevations using the GRID **con** function. As shown in the example, Figure 4, stream lines form strings of adjacent cells, each attributed with its elevation, on a field of cells having the value NODATA. Where gradients are steep each 10-meter cell is a different color, indicating elevation changes of at least one meter between adjacent cells; where gradients are gentle, two or more adjacent cells may show a single color, indicating elevation differences of less than one meter.

<*inventory-grid*> shows the flooded areas and the debris-flow tracks inventoried by fieldwork and air photo interpretation following the June 27, 1995, storm. The ARC command, **polygrid**, was used to create <*inventory-grid*> from the ARC polygon cover of inventory mapping,>. An item, SEL, in the polygon attribute table was assigned integers 1, <*inventory-cover* 2, 3, and 4, to designate nocolor, debris flows, stream floods, and background color, respectively, (1, "nocolor", identifies border zones outside the study area; and 4, "background color", identifies zones within the study area that lack debris-flow or flood features). These integers became the attributes in the .VAT of <*inventory-grid*>. Figures 8 and 9 include <*inventory-cover*> as an overlay, illustrating the distribution of inventoried debris-flows and flooding.

MODEL CONCEPTS

The procedures in the aml can be described as four steps:

STEP 1. Create euclidian allocation grids, based on the elevations of the cells marking the stream lines, for threshold distances of 50m, 30m, and 10m. The **eucallocation** GRID function operates on the grid <*thalweg-elevation-grid*> to extend the area attributed with the elevation of the single-cell-width stream line to distances of 50, 30, and 10 meters, respectively. The resulting grids display the stream lines as bands of cells, 11, 7, and 3 cells wide on a background of NODATA cells. Figures 5a, 5b and 5c show an example using a small part of the Madison County study area.

The **eucallocation** function results in only an approximation of the ideal extension of attributes. Ideally, we would like to extend the elevation of the thalweg cell to adjacent cells in a direction perpendicular to the stream lines for the respective distances, forming a band of cells in which a cross-sectional area of the drainages can be calculated for a given depth. The **eucallocation** function extends a thalweg cell elevation radially, in all directions; but the succession of partial overwrites as grid algebra calculations proceed systematically along rows from the upper left corner of the grid to the lower right corner results in bands of cells that lie across, and approximately perpendicular to, the drainage lines, as shown in Figures 5a, 5b and 5c. (The 10, 30, and 50-meter"expansion distances

are necessary limits to calculations that would otherwise result in spreading the highest elevation over the entire area.) Unfortunately, the systematic sequence of cell-based calculations yields different sequences of partial overwrites for northerly-flowing and southerly-flowing streams. For southerly-flowing streams, expansion of downstream cells having lower elevations partially overwrites the expansion of higher elevation cells upstream. For northerly-flowing streams, expansion of upstream cells having higher elevations partially overwrites the expansion of lower elevation cells downstream.

The depth of the debris-flow surge (user selected, in this case 3 meters), as measured from the thalweg to the surface, is then added to every cell that has a value not equal to NODATA. Subtracting the elevation of the terrain (*<elevation-grid>*) from the elevation of the debris-flow surface yields grids with values representing the depth of the surge. Examples of the results are shown in Figures 6a, 6b and 6c. Because of the upstream-downstream reversal in the sequence of partial overwrites the calculated depths along northerly and southerly flowing streams are not strictly comparable. The potential error from this effect has not yet been fully evaluated, but the reclassification in STEP 1 helps to avoid an inappropriate comparison of depths.

STEP 2. To constrain by stream gradient (slope of thalweg cells) and stream order, begin by creating a euclidian allocation of stream order using a 50-meter distance, the maximum used in Step 1, as illustrated in Figure 7. Then combine the depth grids of Step 1, constrained by slope category and stream order such that where slope is equal or less than 14 degrees and the stream order is greater than 1, the result calls for the surge depth from the 50-meter look; where slope is between 14 and 26 degrees and the stream order is greater than 1, use the surge depth from the 30-meter look; and where the slope is equal to or greater than 26 degrees and stream order is equal to or greater than 1, use the surge depth from the 10-meter look. The result is illustrated by Figure 8.

STEP 3 To remove inventoried areas inundated by the 1995 flood from the gridded hazard area, use the value 3 (for stream floods) from *<inventory-grid>* to assign the value NODATA to areas where debris-flow potential overlaps the mapped flooded area. The result is illustrated in Figure 9. To constrain results to the inventoried study area, use the value 1 from the *<inventory-grid>* to identify areas of NODATA cells in the *<hazard-grid>*.

STEP 4. For ease in plotting and map display, reclassify the result of step 3, above, to show all debris-flow surge depths as one color and convert the grid to a polygon cover, so that it will respond to the ARCPLOT **mapangle** command. The **gridpoly** function results in a cover having a .PAT item named GRID-CODE in which the value for areas affected by the surge is 6, and unaffected areas are 0.

EPILOGUE. The resulting polygon cover contains a few areas showing the effect of a debris-flow surge, even though the areas that contribute to their drainage do not include slopes as steep as 26 degrees. Because debris flows are unlikely to originate from headwater areas having gentle slopes, map units indicating postulated effects of debris flows in the downstream areas were removed by manually editing the polygon cover.

ARC/INFO .AML PROGRAM

The following program was written for specific grids and covers created for the Madison County study. The file names have been generalized to better identify file types. Paths to other directories have been eliminated; therefore, the .aml reads as if all source covers and grids are in the same directory as the .aml. Running requires substantial disk space for making a number of intermediate grids and covers. The study area in Madison County includes an area of 75-100 square kilometers which, at 1:24,000-scale, required a minimum of 40-60 mb of disk capacity. Several expressions in Steps 1 and 3 that could be nested to produce an .aml with fewer command lines, have been left as single expressions to assist in describing the procedure.

Following is the text of the .aml:

```
/* #########  hazmake.aml  #############
/*
/*      Assemble the necessary starting grids
/*              <elevation-grid>
/*              <slope-grid>
/*              <thalweg-elevation-grid>
/*              <stream-order-grid>
/*              <inventory-grid>
/*
/*      STEP 1.  Create euclidian allocation grids, extending the
/*              elevations of the cells marking the stream lines,
/*              <thalweg-elevation-grid>, for threshold distances
/*              of 50m, 30m, and 10m.
/*
display 9999 1
grid
eualloc50 = eucallocation(<thalweg-elevation-grid>, #, #, 50)
eualloc30 = eucallocation(<thalweg-elevation-grid>, #, #, 30)
eualloc10 = eucallocation(<thalweg-elevation-grid>, #, #, 10)
/*
/*              Then add desired debris-flow surge depth to the allocation grid
/*
eual53 = eualloc50 + 3
eual33 = eualloc30 + 3
eual13 = eualloc10 + 3
/*
/*              Next, subtract the terrain elevations from the eual*
/*              elevations, yielding three grids of surge depth over the
/*              digital terrain model.
/*
<flow-depth-grid53> = con((eual53 - <elevation-grid>) gt 0, (eual53 - <elevation-grid>))
<flow-depth-grid33> = con((eual33 - <elevation-grid>) gt 0, (eual33 - <elevation-grid>))
<flow-depth-grid13> = con((eual13 - <elevation-grid>) gt 0, (eual13 - <elevation-grid>))
/*
```

```
/*      STEP 2. To constrain by stream gradient (slope) and stream
/*              order, begin by creating a euclidian allocation of
/*              stream order. Use a 50m look (the max used above).
/*
eucord50 = eucallocation(<stream-order-grid>, #, #, 50)
/*
/*              and create a new grid, constrained by slope category
/*              and stream order, that combines the flow-depth grids.
/*
<first-hazard-grid> =  con(<slope-grid> le 14 && eucord50 gt 1, ~
                                <flow-depth-grid53>, ~
                       con(<slope-grid> lt 26 && <slope-grid> ge 14 && eucord50 gt 1, ~
                                <flow-depth-grid33>, ~
                       con(<slope-grid> ge 26 && eucord50 ge 1, ~
                                <flow-depth-grid13>)))
/*
/*      STEP 3. To remove areas of inventoried floodways from the
/*              gridded hazard area and constrain map to area inventoried,
/*              remember that the <inventory-grid> has the values
/*              1, 2, 3, and 4, = nocolor, debris flows, stream floods,
/*              and background color, respectively.
/*
<second-hazard-grid> = setnull (<inventory-grid> == 3, <first-hazard grid>)
/*
/*              and to remove features outside the study area
/*
<third-hazard-grid> = setnull (<inventory-grid> == 1, <second-hazard-grid>)
/*
/*      STEP 4. For ease in plotting and map display, reclassify
/*              from flood depths to show all one color
/*
<final-hazard-grid> = con (<third-hazard-grid> gt 0, 6)
/*
/*              and convert the grid to a polygon cover, so that it
/*              will respond to the arcplot 'mapangle' command.
/*
<hazard-cover> = gridpoly(<final-hazard-grid>)
/*
/*              which creates a cover with an item named GRID-CODE in
/*              the .PAT that has the value 6.
/*
/*      then leave grid
/*
quit
```

```
/*      and in ARC, build the cover for lines
/*
build <hazard-cover> line
/*
/*              (Apparently, 'gridpoly' creates a .PAT, but does not
/*              automatically create arc topology. So to plot the lines,
/*              it is necessary to build line topology.)
&return
```

DISCUSSION

The hazard map (Morgan and others, 1997, plate 2) was prepared using the foregoing .aml. For a preliminary evaluation of performance, it can be compared with the inventory of actual debris-flow hazard areas for the storm of June 27, 1995, as shown in Table 1. In the Madison County study area, which includes 2,680,058 10-meter cells (26,801 hectares), the model delineates areas totaling 400,253 cells as having a potential for debris-flow hazard, of which 36,066 (9.0%) were affected by inventoried debris-flows from the storm of June 27, 1995. Much of the over prediction can be attributed to the decrease in rainfall with distance away from the center of the "bullseye" formed by the storm isohyets (see Morgan and others, 1997, fig 1.). As should be expected, restricting the total area to progressively higher storm rainfall -- the areas included within the 350mm, 450mm, 550mm, and 650mm isohyets, respectively -- increases the ratio of inventoried to model-predicted (expressed as percent) to 11.0%, 15.8%, 21.2%, and 26.0%, respectively.

Of the 2,279,805 cells delineated as outside the zone of potential hazard, 2,253,019 cells (98.8%) had no inventoried hazard and 26,786 cells (1.2%) had inventoried debris-flow effects. Visually, the hazard map appears to discriminate satisfactorily between areas of potential debris-flow hazard and areas unlikely to be at hazard from debris flows.

On the disappointing side, of a total of 62,852 inventoried hazard cells, the model predicts a potential hazard in 36,066 cells, or about 57%, leaving 26,768 cells, approximately 43% of the inventoried hazard cells, in areas where the potential for hazard is not predicted by the model. These proportions remain constant, even when areas are restricted to the limits delineated by the 350mm, 450mm, 550mm, and 650mm isohyets for the storm rainfall. Some of this error may be inherent in the use of a single average depth, applied to all stream channels, because volumes (and associated depths) may be expected to vary from one drainage to another. However, a major part of the error can be attributed to the inability of the digital terrain model to discriminate the correct courses of drainage channels across fans and other areas having slopes of 14 degrees and less. As shown on Figure 3, several channels delineated in the digital terrain model by GRID functions, depart from courses that human interpretation from the contours might select. These departures influence succeeding grids, and comparison of Figure 3 with Figures 8 and 9 shows that to be a major source of error. It seems clear that significant reductions in error would result from comparing the stream-net grid with the contour map, and editing the grid to conform with stream lines interpreted from the contours prior to preparation of the *<stream-order-grid>* and the *<thalweg-elevation-grid>*.

Table 1. a. Comparing model-predicted hazard cell count (predicted frequency) with inventoried hazard cell count (actual frequency).

	Predicted Hazard	Predicted No-hazard	Total
Inventoried hazard	36,066	26,786	62,852
Inventoried no-hazard	364,187	2,253,019	2,617,206
Total	400,253	2,279,805	2,680,058

Table 1. b. Comparing "successful" predictions of hazard cells with "failed" (incorrect) model predictions.

	Predicted Hazard	Predicted No-hazard	Total
Successes*	36,066	2,253,019	2,289,085
Failures*	364,187	26,786	390,973
Total	400,253	2,279,805	2,680,058

*("Success" = Predicted and Inventoried results the same.)
*("Failure" = Predicted and inventoried results are different.)

Yates continuity correction, on the null hypothesis that the inventoried and predicted distributions are not the same,

400,253 x 62,852/2,680,058 = 9,386.6;

which is less than 36,066.

Table 1. Comparisons of model-predicted and inventoried hazards. Cell counts in the Madison County study area, approximately bounded by the 350 mm isohyet of the June 27, 1995 storm (see Wieczorek and others, in prep, fig. 2a), which also approximates the limits of debris-flow events associated with the storm (see Wieczorek and others, 1996, pl. 1).

The frequency of erroneous predictions decreases with increasing slope angles, as shown by the following tabulation:

Slope (deg)	Erroneous predictions (percent)
≤ 14	42-52
14-26	30-32
26-34	16-21
34-45	2-5
> 45	< 1

The range in erroneous predictions is affected by the severity of the storm rainfall. Over 42% of the erroneous predictions where slopes are 14 degrees or less, occur within the 350mm isohyet, and this proportion reaches 52% for the area within the 650mm isohyet for storm rainfall.

The associations of erroneous predictions with gentle slope areas indicates that improved map accuracy should result if the elevation grid, which forms the basis for the derivative maps of slope, stream lines, and stream order, were more accurate and more detailed than is presently provided by the contour DLG's, at least in low-slope (less than 14 degrees) areas. Perhaps supplemental digital elevation data for low-slope areas could be incorporated into the elevation grid. The results of testing indicate that, although the combination of available terrain data and GIS procedures in this particular .aml produces too many errors to be a reliable cell-specific indicator of potential hazard, it may be useful in providing a general guide to areas where special mitigation or emergency-response measures should be considered.

Improved performance can be expected from more detailed digital terrain model. Editing the stream-net grid to conform with visual picks of drainage lines from the contour maps, and to identify streams that lack steep slopes (26 degrees and greater) in their watershed areas, can be expected to reduce errors. Extending application to other areas will surely require adjustments in:

Step 1. user-designated factors such as surge depth and limiting distances for euclidian allocations; development and incorporation of subroutines to compensate for the effect of upstream and downstream calculation sequences, should be investigated,

Step 2. adjustments to slope and stream order constraints, and addition of other possible constraints, in the way depth grids are combined,

Step 3. map input, for example, mapped flood plains and debris-flow fans could substitute for actual post-storm inventories,

Step 4. a routine that results in more accurate depth grids in Step 1 might permit the use of model depths instead of restricting the model to a "yes" or a "no" on inundation.

ACKNOWLEDGMENTS

The concept of modeling the potential for debris-flow hazard with a user-specified surge height should be attributed to B. A. Morgan and G. F. Wieczorek, both of the U.S. Geological Survey in Reston, VA. The development of the .aml was greatly aided by discussions with John Jones, Susan Price, David Soller, and Patricia Packard, all of the U.S. Geological Survey in Reston, VA. We are also grateful to Rachael Hauser, U.S. Geological Survey and University of California, Riverside, and to Gerald Wieczorek, U.S. Geological Survey, Reston, VA, for helpful reviews of the manuscript.

REFERENCES

ESRI (Environmental Systems Research Institute), 1992, Hydrologic Modeling Tools: Environmental Systems Research Institute Inc., Redlands, California, p. 1-1 - 1-14.

ESRI (Environmental Systems Research Institute), 1991, Cell-based modeling with GRID: Environmental Systems Research Institute Inc., Redlands, California, 333 p.

Ellen, S.D., Mark, R.K., Cannon, S.H., and Knifong, D.L, 1993, Map of debris-flow hazard in the Honolulu District of Oahu, Hawaii: U.S. Geological Survey Open-File Report OF 93-213, 25 p., plate 1:30,000.

Iverson, R.M., Schilling, S.P., and Vallance, J.W., 1998, Objective delineation of lahar-inundation hazard zones: Geological Society of America Bulletin, v. 10, no. 8, p. 972-984.

Morgan, B.A., Wieczorek, G.F., and Campbell, R.H., in press, Historical and potential debris-flow hazard map of area affected by the June 27, 1995, storm in Madison County, Virginia: U.S. Geological Survey Miscellaneous Investigation Series Map I-2623B, 1:24,000.

Morgan, B.A., Wieczorek, G.F., Campbell, R.H., and Gori, P.H., 1997, Debris-flow hazards in areas affected by the June 27, 1995, storm in Madison county, Virginia: U.S. Geological Survey Open-File Report 97-438, 15 p., 2 tables, 2 plates 1:24,000.

Schilling, S.P., 1998, LAHARZ: GIS programs for automated delineation of lahar hazard zones. U.S. Geological Survey Open-file Report 98-638, 80 p.

Wieczorek, G.F., Morgan, B.A., Campbell, R.H., Orndorff, R.C., Burton, W.C., Southworth, C.S., and Smith, J.A., 1996, Preliminary inventory of debris-flow and flooding effects of the June 27, 1995, storm in Madison County, Virginia, showing time sequence of positions of storm-cell center: U.S. Geological Survey Open-File Report 96-13, 8 p., 1 plate 1:24,000.

Wieczorek, G.F., Morgan, B.A., and Campbell, R.H., (in prep.), Debris-flow hazards in the Blue Ridge of central Virginia: Environmental and Engineering Geosciences Journal, 65 ms. p.

Examples of input data and grid manipulations to produce a preliminary debris-flow hazards map.



Scale and north arrow for all figures.



Figure 1 Contours, streams and roads from digital line graph data, on shaded relief map from digital terrain model. (10-m cells)



Figure 2. Slope map from digital terrain model.



Figure 3. Stream order derived from digital terrain model. Circled areas show model-defined channels that do not follow the courses expected from examination of the contours.



Figure 4. Elevations of 10-m cells marking stream lines derived from digital terrain model.



Figure 5a. 10 meter euclidian allocation of stream elevations, with 3 meter depth added.



Figure 5b. 30 meter euclidian allocation of stream elevations, with 3 meter depth added.



Figure 5c. 50 meter euclidian allocation of stream elevations, with 3 meter depth added.



Figure 6a. Depths derived by subtracting terrain model (gridded elevations) from grid of Fig. 5a, subject to the constraint that depth cannot be negative.



Figure 6b. Depths derived by subtracting terrain model from grid of Fig. 5b, subject to the constraint that depth is not negative.



Figure 6c. Depths derived by subtracting terrain model from grid of Fig. 5c, Subject to the constraint that depth is not negative.



Figure 7. 50 meter euclidian allocation of stream orders (see Fig. 3).



Figure 8. Combining grids of Figures 6a, 6b and 6c with constraints described in text, showing overlay of mapped debris flow and flood extent from events associated with the storm of June 27, 1995.



Figure 9. Showing effects of removing mapped flooding from debris-flow hazard areas identified by model.